Pensieve header: The zero-co Adjoint representation.

The Adjoint action

```
DistinctQ[is___] := (Sort[{is}] === Union[{is}]);
expr_ // Ad[t_. a[i_, j_]] := Expand[expr /. {
    a[k_, j] /; DistinctQ[i, j, k] :> e^(-t b_i) a[k, j] + b_k/b_i (1 - e^(-t b_i)) a[i, j],
    a[j, k_] /; DistinctQ[i, j, k] :> e^(t b_i) a[j, k] + b_j/b_i (1 - e^(t b_i)) a[i, k],
    a[k_, i] /; DistinctQ[i, j, k] :>
     a[k, i] + (1 - e^(-t b_i)) a[k, j] + b_k/b_i (e^(-t b_i) - 1) a[i, j],
    a[j, i] :> e^(t b_i) a[j, i] + b_j/b_i (1 - e^(t b_i)) a[i, j],
    a[k_, l_] /; DistinctQ[i, j, k, l] :> a[k, l],
    a[i, k_] :> a[i, k],
    a[k_, k_] :> a[k, k],
    a[k_, l_] :> TBD[a[i, j], a[k, l]]
   }]
```

```
Table[a[i, j] → SeriesCoefficient[a[i, j] // Ad[ϵ a[1, 2]], {ϵ, 0, 1}] // Simplify,
  {i, 3}, {j, 3}] // MatrixForm
```

$$\begin{pmatrix} a[1,1] \to 0 & a[1,2] \to 0 & a[1,3] \to 0 \\ a[2,1] \to a[2,1]\, b_1 - a[1,2]\, b_2 & a[2,2] \to 0 & a[2,3] \to a[2,3]\, b_1 - a[1 \\ a[3,1] \to a[3,2]\, b_1 - a[1,2]\, b_3 & a[3,2] \to -a[3,2]\, b_1 + a[1,2]\, b_3 & a[3,3] \to 0 \end{pmatrix}$$

The semigroup property

```
Table[a[i, j], {i, 3}, {j, 3}] // Ad[x a[1, 2]] // Ad[y a[1, 2]] // FullSimplify //
 MatrixForm
```

$$\begin{pmatrix} a[1,1] & a[1,2] & \\ \frac{a[1,2]\, b_2 + e^{(x+y)\, b_1} (a[2,1]\, b_1 - a[1,2]\, b_2)}{b_1} & a[2,2] & \frac{a[1,3]\, b_2 + e}{} \\ a[3,1] + \left(1 - e^{-(x+y)\, b_1}\right) a[3,2] + \frac{\left(-1 + e^{-(x+y)\, b_1}\right) a[1,2]\, b_3}{b_1} & \frac{e^{-(x+y)\, b_1} \left(a[3,2]\, b_1 + \left(-1 + e^{(x+y)\, b_1}\right) a[1,2]\, b_3\right)}{b_1} & \end{pmatrix}$$

```
Table[a[i, j], {i, 3}, {j, 3}] // Ad[(x + y) a[1, 2]] // FullSimplify // MatrixForm
```

$$\begin{pmatrix} a[1,1] & a[1,2] & a[1, \\ \frac{a[1,2]\, b_2 + e^{(x+y)\, b_1} (a[2,1]\, b_1 - a[1,2]\, b_2)}{b_1} & a[2,2] & \frac{a[1,3]\, b_2 + e^{(x+y)\, b_1} (a[2,}{b_1} \\ \frac{\left(a[3,1] + \left(1 - e^{-(x+y)\, b_1}\right) a[3,2]\right) b_1 + \left(-1 + e^{-(x+y)\, b_1}\right) a[1,2]\, b_3}{b_1} & \frac{e^{-(x+y)\, b_1} \left(a[3,2]\, b_1 + \left(-1 + e^{(x+y)\, b_1}\right) a[1,2]\, b_3\right)}{b_1} & a[3, \end{pmatrix}$$

```
Table[a[i, j] → (
    (a[i, j] // Ad[x a[1, 2]] // Ad[y a[1, 2]]) ==
      (a[i, j] // Ad[(x + y) a[1, 2]]) // Simplify
   ), {i, 3}, {j, 3}] // MatrixForm
```

a[1, 1] → True	a[1, 2] → True	a[1, 3] → True
a[2, 1] → True	a[2, 2] → True	a[2, 3] → True
a[3, 1] → True	a[3, 2] → True	a[3, 3] → True

Verifying R3

```
(R3L = Table[a[i, j], {i, 4}, {j, 4}] // Ad[a[1, 2]] // Ad[a[1, 3]] // Ad[a[2, 3]] //
    Simplify) // MatrixForm
```

$$\begin{pmatrix} a[1,1] \\ \frac{(-a[2,3]+e^{b_1}(a[2,1]+a[2,3]))\,b_1-(-1+e^{b_1})(a[1,2]+a[1,3])\,b_2}{b_1} \\ \frac{-(-1+e^{b_1})(a[1,2]+a[1,3])\,b_2\,b_3+b_1\,(e^{b_2}(-a[3,2]+e^{b_1}(a[3,1]+a[3,2]))\,b_2-(-1+e^{b_2})(-a[2,3]+e^{b_1}(a[2,1]+a[2,3]))\,b_3)}{b_1\,b_2} \\ \frac{e^{-b_1}((-a[4,2]-a[4,3]+e^{b_1}(a[4,1]+a[4,2]+a[4,3]))\,b_1-(-1+e^{b_1})(a[1,2]+a[1,3])\,b_4)}{b_1} \quad \frac{e^{-b_1-b_2}\,(}{} \end{pmatrix}$$

```
(R3R = Table[a[i, j], {i, 4}, {j, 4}] // Ad[a[2, 3]] // Ad[a[1, 3]] // Ad[a[1, 2]] //
    Simplify) // MatrixForm
```

$$\begin{pmatrix} a[1,1] \\ \frac{(-a[2,3]+e^{b_1}(a[2,1]+a[2,3]))\,b_1-(-1+e^{b_1})(a[1,2]+a[1,3])\,b_2}{b_1} \\ \frac{-(-1+e^{b_1})(a[1,2]+a[1,3])\,b_2\,b_3+b_1\,(e^{b_2}(-a[3,2]+e^{b_1}(a[3,1]+a[3,2]))\,b_2-(-1+e^{b_2})(-a[2,3]+e^{b_1}(a[2,1]+a[2,3]))\,b_3)}{b_1\,b_2} \\ \frac{e^{-b_1}((-a[4,2]-a[4,3]+e^{b_1}(a[4,1]+a[4,2]+a[4,3]))\,b_1-(-1+e^{b_1})(a[1,2]+a[1,3])\,b_4)}{b_1} \quad \frac{e^{-b_1-b_2}\,(}{} \end{pmatrix}$$

```
Table[a[i, j] → (
    (a[i, j] // Ad[a[1, 2]] // Ad[a[1, 3]] // Ad[a[2, 3]]) ==
      (a[i, j] // Ad[a[2, 3]] // Ad[a[1, 3]] // Ad[a[1, 2]]) // Simplify
   ), {i, 4}, {j, 4}] // MatrixForm
```

a[1, 1] → True	a[1, 2] → True	a[1, 3] → True	a[1, 4] → True
a[2, 1] → True	a[2, 2] → True	a[2, 3] → True	a[2, 4] → True
a[3, 1] → True	a[3, 2] → True	a[3, 3] → True	a[3, 4] → True
a[4, 1] → True	a[4, 2] → True	a[4, 3] → True	a[4, 4] → True